Dataplot Vol 2 Vol 1

WELCH SATTERTHWAITE

Name:
GUM WELCH SATTERTHWAITE (LET)
VARIANCES WELCH SATTERTHWAITE (LET)
Type:
Let Subcommand
Purpose:
Compute effective degrees of freedm based on the Welch-Satterthwaite equation.
Description:
The Welch-Saitterthwaite equation is used to calculate the approximate degrees of freedom for a linear combination of independent sample variances. If the various variances are not independent, then the Welch-Satterthwaite approximation may not be valid.

Dataplot supports two versions of the Welch-Saitterthwaite approximation.

1. In performing uncertainty analysis, type A (random) and type B (systematic) errors often need to be combined. One issue in combining these uncertainties is how to determine the effective degrees of freedom.

Degrees of freedom for type A uncertainties are the degrees of freedom for the repspective type A standard deviations. Degrees of freedom for type B uncertainties are often available from published reports, certificates, or they may be supplied by a vendor. In cases where a type B component is being provided by scientific judgement or the degrees of freedom are otherwise not known, it is typical to assume infinite degrees of freedom (for practical purposes, infinite degrees of freedom may simply be given as a large value such as 10,000).

The Welch-Satterthwaite approximation for the degrees of freedom for the standard uncertainty is

$$\nu = \frac{u^{4}} {\sum_{i=1}^{k} {\frac{a_{i}^{4} s_{i}^{4}} {\nu_{i}}}}$$

where

 k = the number of components si = the standard deviation of the i-th component $$\nu_{i}$$ = the standard deviation of the i-th component ai = the sensitivity coefficient of the i-th component u = the standard uncertainty = $$\sqrt{\sum_{i=1}^{k}{a_{i}^{2} s_{i}^{2}}}$$

For this command, the si, νi, and ai are given as inputs and u will be computed from the ai and si components.

The sensitivity coefficients are derived from partial derivatives of the measurement equation. For the case of additive, independent uncertainties, these can often be set to 1.

The NIST/SEMATACH e-Handbook of Statistical Methods gives some examples of this. In particular, it shows some examples of determining the sensitivity coefficients.

2. Given n variances, $$s_{i}^{2}$$, with their associated degrees of freedom, νi, we want to compute the pooled standard deviation and the associated degrees of freedom for the linear combination

$$\sum_{i=1}^{n}{k_{i} s_{i}^{2}}$$

where ki is typically 1/(νi + 1). The Welch-Saitterwaithe approximation for the effective degrees of freedom is given by

$$\frac{\left( \sum_{i=1}^{n}{k_{i} s_{i}^{2}} \right) ^{2}} {\sum_{i=1}^{n}{\frac{(k_{i} s_{i}^{2})^{2}} {\nu_{i}}} }$$

A pooled standard deviation is then computed as

$$s_{\mbox{pooled}} = \sqrt{ \frac{\sum_{i=1}^{n}{(\nu_{i} - 1) s_{i}^{2}}} {\sum_{i=1}^{n}{\nu_{i} - 1} }}$$
Syntax 1:
LET <df> = GUM WELCH SATTERTHWAITE <ysd> <ydf> <ysens>
where <ysd> is a variable containing the standard deviatins for the various components of uncertainty;
<ydf> is a variable containing the degrees of freedom for the various components of uncertainty;
<ysens> is a variable containing the sensitivity coefficients for the various components of uncertainty;
and where <df> is a parameter where the computed degrees of freedom is saved.
Syntax 2:
LET <df> <poolsd> = VARIANCES WELCH SATTERTHWAITE <yvar> <ydf>
where <yvar> is a variable containing the variances for the various components of uncertainty;
<ydf> is a variable containing the degrees of freedom for the various components of uncertainty;
<df> is a parameter where the computed degrees of freedom is saved;
and where <poolsd> is a parameter where the computed pooled standard deviation is saved.
Examples:
LET IDF = GUM WELCH SATTERTHWAITE YSD YDF YA
LET IDF POOLSD = VARIANCES WELCH SATTERTHWAITE YVAR YDF
Default:
None
Synonyms:
None
Related Commands:
 T TEST = Perform a two sample t-test. CONSENSUS MEANS = Compute a consensus mean and its associated uncertainty.
Reference:
Satterthwaite (1946), "An Approximate Distribution of Variance Components", Biometrics Bulletin, 2: 110-114.

Welch (1947), "The Generalization of Students's Problem when Several Different Population Variances are Involved", Biometrika, 34: 28-35.

"Guide to the Expression of Uncertainty in Measurement", ISO, Geneva (1993).

"NIST/SEMATECH Handbook of Statistical Methods", Measurement Process Characterization chapter, " http://www.itl.nist.gov/div898/handbook/mpc/mpc.htm", June, 2003.

Applications:
Uncertainty Analysis
Implementation Date:
2017/01
2017/07: Updated the formula for the pooled standard deviation
Program 1:

SKIP 25
LET N1 = SIZE Y1
LET NU1 = N1 - 1
LET VAR1 = VARIANCE Y1
LET N2 = SIZE Y2
LET NU2 = N2 - 1
LET VAR2 = VARIANCE Y2
LET YVAR = DATA VAR1 VAR2
LET YDF = DATA NU1 NU2
LET DF POOLSD = VARIANCES WELCH SATTERTHWAITE YVAR YDF
LET DF = ROUND(DF,2)
LET POOLSD = ROUND(POOLSD,2)
PRINT "Degrees of Freedom:   ^DF"
PRINT "Pooled SD:            ^POOLSD"

The following output is generated

Degrees of Freedom:   248.09
Pooled SD:            339.52

Program 2:

LET YSD = DATA 0.00371 0.00191  0.00191 0.00006
LET YDF = DATA 2 1000 1000 1000
LET YA  = DATA 1 1 1 1
LET DF = GUM WELCH SATTERTHWAITE YSD YDF YA
LET DF = ROUND(DF,2)
PRINT "Degrees of Freedom:   ^DF"

The following output is generated

Degrees of Freedom:   4.68


NIST is an agency of the U.S. Commerce Department.

Date created: 07/20/2017
Last updated: 07/20/2017